

Упрощение формул Excel путем именования фрагментов с помощью функции LET

Функция LET появилась в версии Office 365 (по подписке) и будет доступна, начиная с версии Office 2021. Функция LET позволяет присваивать имена фрагментам формулы, а затем использовать эти имена в вычислениях. Формулы становятся короче и лучше читаемы. Это похоже на использование имен ячеек, диапазонов и формул в диспетчере имен. Но имена, присвоенные функцией LET, действуют только в пределах указанной формулы, и нигде больше. При использовании в русском Office имя функции сохраняется (не переводится на русский язык).

Синтаксис




Рис. 1. Синтаксис функции LET

Более строго, синтаксис функции LET:

LET(имя1; значение_имени1; [имя2; значение_имени2]; ...; вычисление)

Имя1 – обязательный аргумент, должен начинаться с буквы или нижнего подчеркивания.

Значение_имени1 – обязательный аргумент, содержит значение для Имя1.

имя2/значение_имени2 – необязательные аргументы; всего может быть не более 126 таких пар.

Вычисление – обязательный аргумент, формула, использующая имена.

В простейшем случае:




Рис. 2. Возведение в степень с помощью функции LET


Внутри функции LET определено имя x и его значение 3, а также имя y со значением 4. С этими именами выполняется действие x^y . Этот пример поясняет базовые возможности функции LET. Для подобных простых вычислений использование LET не оправдано. Поэтому обратимся к более сложным формулам, в которых использование LET будет полезным.

Функция LET устраняет дублирование фрагментов формулы

Допустим, вам нужно вывести все рабочие дни между двумя датами, хранящимися в ячейках листа. Функция LET присвоит имя исходному массиву дат, а затем использует это имя в функциях ФИЛЬТР() и ДЕНЬНЕД():

```
=LET(даты;ПОСЛЕД(C5-C4+1;;C4);ФИЛЬТР(даты;ДЕНЬНЕД(даты;2)<6))
```

Если вы не знакомы с функциями динамических массивов, вам для начала [сюда](#))



	A	B	C	D	E	F	G
1	07.07.2021						
2	08.07.2021						
3	09.07.2021						
4	12.07.2021		07.07.2021				
5	13.07.2021		28.07.2021				
6	14.07.2021						
7	15.07.2021						
8	16.07.2021						
9	19.07.2021						
10	20.07.2021						
11	21.07.2021						
12	22.07.2021						
13	23.07.2021						
14	26.07.2021						
15	27.07.2021						
16	28.07.2021						

Рис. 3. Упрощение формулы за счет именованного ее фрагмента

Первый аргумент функции LET – даты – вводит имя внутри формулы. Вторым аргументом функции LET задается массив дат: ПОСЛЕД(C5-C4+1;;C4). Этот массив включает на одно значение больше, чем разность в ячейках C5 и C4, начинается с C4 с шагом 1.

Третьим аргументом функции LET осуществляется вычисление с массивом дат:

ФИЛЬТР(даты;ДЕНЬНЕД(даты;2)<6). А именно, фильтрует массив дат, оставляя только даты, для которых ДЕНЬНЕД() меньше 6, т.е., исключает выходные дни.

Без функции LET формула могла выглядеть так:

```
=ФИЛЬТР(ПОСЛЕД(C5-C4+1;;C4);ДЕНЬНЕД(ПОСЛЕД(C5-C4+1;;C4);2)<6)
```

Объявляя переменные и присваивая им значения, функция LET упрощает написание сложных формул. Внутри LET может использоваться несколько пар: *имя/значение*. Аргумент *вычисление* всегда будет последним в формуле.

Функция LET дает три существенных преимущества

Простота. Использование имен для повторяющихся фрагментов упрощает формулы и облегчает их редактирование. Если требуется, вы вносите одно изменение, а не несколько. Это быстрее и снижает вероятность ошибки.

Ясность. Использование говорящих имен облегчает понимание того, как работает формула. Человеку, написавшему формулу и использующему её, гораздо проще дешифровать внутреннюю структуру записи.

Производительность. Удаление дублирования сокращает время вычислений. Это незаметно в обычных книгах Excel, но очень важно в моделях с миллионами записей.

Использование LET во вложенных конструкциях ЕСЛИ

Допустим у вас есть результаты трех тестов по трем отдельным предметам (столбцы В, С и D). Вам нужно рассчитать среднее значение для каждого человека и выставить оценку в соответствии с критерием в правой таблице:

	A	B	C	D	E	F	G	H
1	Фамилия	Математика	Физика	Химия			Оценка	Критерий
2	Смирнов	260	248	275			Отл.	250 и более
3	Иванов	229	190	224			Хор.	200-249
4	Кузнецов	152	270	185			Уд.	160-199
5	Соколов	207	272	288			Неуд.	159 и менее
6	Попов	188	169	165				
7	Лебедев	190	289	279				
8	Козлов	274	227	181				
9	Новиков	178	181	222				
10	Морозов	226	271	286				
11	Петров	180	262	258				
12	Волков	271	157	236				
13	Соловьёв	250	232	152				

Рис. 4. Результаты тестов для оценки

Без функции LET вы, скорее всего, выставите оценку с помощью вложенных ЕСЛИ:

```
=ЕСЛИ(
    СРЗНАЧ(В2:D2)>=250;"Отл.";
    ЕСЛИ(
        СРЗНАЧ(В2:D2)>=200;"Хор.";
        ЕСЛИ(СРЗНАЧ(В2:D2)>=160;"Уд. ";"Неуд.")
    )
)
```

=ЕСЛИ(СРЗНАЧ(В2:D2)>=250;"Отл. ";ЕСЛИ(СРЗНАЧ(В2:D2)>=200;"Хор. ";ЕСЛИ(СРЗНАЧ(В2:D2)>=160;"Уд. ";"Неуд. ")))

	A	B	C	D	E	F	G	H	I	J
1	Фамилия	Математика	Физика	Химия	Оценка		Оценка	Критерий		
2	Смирнов	260	248	275	Отл.		Отл.	250 и более		
3	Иванов	229	190	224	Хор.		Хор.	200-249		
4	Кузнецов	152	270	185	Хор.		Уд.	160-199		
5	Соколов	207	272	288	Отл.		Неуд.	159 и менее		
6	Попов	188	169	165	Уд.					
7	Лебедев	190	289	279	Отл.					
8	Козлов	274	227	181	Хор.					
9	Новиков	178	181	222	Уд.					
10	Морозов	226	271	286	Отл.					
11	Петров	180	262	258	Хор.					
12	Волков	271	157	236	Хор.					
13	Соловьёв	250	232	152	Хор.					

Рис. 5. Оценка с помощью функций ЕСЛИ

Проблема в том, что СРЗНАЧ(В2:D2) рассчитывается трижды. Чтобы сделать формулу более эффективной (и наглядной) мы можем присвоить среднему значению имя:

```
=LET(
    среднее;СРЗНАЧ(В2:D2);
    ЕСЛИ(среднее>=250;"Отл. ";
        ЕСЛИ(среднее>=200;"Хор. ";
            ЕСЛИ(среднее>=160;"Уд. ";"Неуд. ")
        )
    )
```

)
)
)

=LET(среднее;СРЗНАЧ(B2:D2);ЕСЛИ(среднее>=250;"Отл.";ЕСЛИ(среднее>=200;"Хор.";ЕСЛИ(среднее>=160;"Уд."; "Неуд.")))

	A	B	C	D	E	F	G	H	I	J	K
1	Фамилия	Математика	Физика	Химия	Оценка		Оценка	Критерий			
2	Смирнов	260	248	275	Отл.		Отл.	250 и более			
3	Иванов	229	190	224	Хор.		Хор.	200-249			
4	Кузнецов	152	270	185	Хор.		Уд.	160-199			
5	Соколов	207	272	288	Отл.		Неуд.	159 и менее			
6	Попов	188	169	165	Уд.						
7	Лебедев	190	289	279	Отл.						
8	Козлов	274	227	181	Хор.						
9	Новиков	178	181	222	Уд.						
10	Морозов	226	271	286	Отл.						
11	Петров	180	262	258	Хор.						
12	Волков	271	157	236	Хор.						
13	Соловьёв	250	232	152	Хор.						

Рис. 6. Выставление оценки с использованием LET

Если вам понадобится добавить еще один экзамен, то вы поправите формулу лишь в одном месте:

=LET(среднее;СРЗНАЧ(B2:E2);ЕСЛИ(среднее>=250;"Отл.";ЕСЛИ(среднее>=200;"Хор.";ЕСЛИ(среднее>=160;"Уд."; "Неуд.")))

	A	B	C	D	E	F	G	H	I	J
1	Фамилия	Математика	Физика	Химия	Биология	Оценка		Оценка	Критерий	
2	Смирнов	260	248	275	208	Хор.		Отл.	250 и более	
3	Иванов	229	190	224	243	Хор.		Хор.	200-249	
4	Кузнецов	152	270	185	166	Уд.		Уд.	160-199	
5	Соколов	207	272	288	276	Отл.		Неуд.	159 и менее	
6	Попов	188	169	165	290	Хор.				
7	Лебедев	190	289	279	274	Отл.				
8	Козлов	274	227	181	142	Хор.				
9	Новиков	178	181	222	150	Уд.				
10	Морозов	226	271	286	230	Отл.				
11	Петров	180	262	258	232	Хор.				
12	Волков	271	157	236	191	Хор.				
13	Соловьёв	250	232	152	283	Хор.				

Рис. 7. Добавление еще одного экзамена требует лишь одного изменения в формуле

Без LET вам пришлось бы вносить изменения в трех местах.

Использование нескольких имен в функции LET

Допустим у вас есть ФИО, и нужно извлечь имя. Следующая формула справляется с этой задачей, но понять, как она работает совсем не просто:

=ПСТР(A2;ПОИСК(" ";A2)+1;ПОИСК(" ";A2;ПОИСК(" ";A2)+1)-ПОИСК(" ";A2)-1)

	A	B	C	D	E
1	ФИО	Имя			
2	Иванов Алексей Леонидович	Алексей			
3	Смирнов Артём Максимович	Артём			
4	Кузнецов Вадим Матвеевич	Вадим			
5	Попов Владимир Никитич	Владимир			
6	Васильев Валентин Олегович	Валентин			
7	Петров Данил Павлович	Данил			
8	Соколов Денис Пётрович	Денис			
9	Михайлов Дмитрий Романович	Дмитрий			
10	Новиков Егор Сергеевич	Егор			
11	Фёдоров Кирилл Станиславович	Кирилл			

Рис. 8. Извлечение имени из ФИО

Чтобы работа формулы стала более понятной, можно ввести несколько имен:

- ФИО = A2
- пробел1 = ПОИСК(" ";A2)
- пробел2 = ПОИСК(" ";A2;ПОИСК(" ";A2)+1)

Теперь формула стала намного понятнее:

```
=LET(
    ФИО;A2;
    пробел1;ПОИСК(" ";A2);
    пробел2;ПОИСК(" ";A2;ПОИСК(" ";A2)+1);
    ПСТР(ФИО;пробел1+1;пробел2-пробел1-1)
)
```

	A	B	C	D	E	F	G
1	ФИО	Имя					
2	Иванов Алексей Леонидович	Алексей					
3	Смирнов Артём Максимович	Артём					
4	Кузнецов Вадим Матвеевич	Вадим					
5	Попов Владимир Никитич	Владимир					
6	Васильев Валентин Олегович	Валентин					
7	Петров Данил Павлович	Данил					
8	Соколов Денис Пётрович	Денис					
9	Михайлов Дмитрий Романович	Дмитрий					
10	Новиков Егор Сергеевич	Егор					
11	Фёдоров Кирилл Станиславович	Кирилл					

Рис. 9. Та же формула, но с использованием имен

Функция LET и динамические массивы

В предыдущих примерах формула с LET вводилась в одну ячейку, а далее копировалась вдоль столбца. Чаще у нас на входе будет массив, и результат также будет представлять из себя массив. В Excel 365 обработать такие ситуации можно с помощью одной формулы с поддержкой динамических массивов.

Предположим, в одной таблице у вас есть график экзаменов (см. рис. 10, диапазон A2:C21), а во второй таблице вы хотите вывести только ту часть графика, которая приходится на n ближайших дней, не включая сегодняшний. Да к тому же отсортировать список по фамилиям по алфавиту.

Определите имена внутри функции LET, и используйте функции динамических массивов СОРТ и ФИЛЬТР. Результатом будет массив записей в диапазоне E6:G10.

```
=LET(
  данные;$A$2:$C$21;
  даты;C2#;
  сегодня;E1;
  n;F3;
  СОРТ(
    ФИЛЬТР(
      данные;
      (даты>сегодня)*(даты<=сегодня+n);
      "Нет экзаменов"
    )
  )
)
```

Студент	Предмет	Дата экзамена
Мельникова	Математика	14.08.2021
Иванова	Английский	15.08.2021
Буракшаева	Биология	12.08.2021
Фурсова	Искусство	14.08.2021
Сапсай	Английский	19.08.2021
Богословский	Химия	16.08.2021
Самбикина	Математика	16.08.2021
Шпак	Биология	12.08.2021
Пименов	Английский	20.08.2021
Сигида	Химия	19.08.2021
Миронова	Биология	18.08.2021
Безуглова	Математика	22.08.2021
Сергеева	Английский	13.08.2021
Перфильева	Искусство	22.08.2021
Химич	Математика	16.08.2021
Бондина	Математика	12.08.2021
Лебедева	Английский	12.08.2021
Мощева	Химия	14.08.2021
Моругина	Биология	19.08.2021
Прокопенко	Искусство	16.08.2021

Рис. 10. Использование LET с динамическими массивами

Поскольку даты экзаменов я задал в ячейке C2 случайной функцией СЛМАССИВ, нажимая F9 вы будете изменять даты в C2:C21. Соответственно будет меняться таблица, начинающаяся в E6.

Извлечение предпоследнего слова из текстовой строки

Допустим у вас есть серия фраз, из которых нужно извлечь предпоследнее слово. В принципе понятно, как найти решение, но также ясно, что оно будет не простым, и основано на целом ряде шагов. Можно, например:

- 1) сжать пробелы (вдруг между какими-то словами используется более одного пробела);
- 2) подсчитать длину исходной фразы;
- 3) длину фразы с удаленными пробелами;
- 4) узнать количество слов во фразе;
- 5) заменить пробел перед предпоследним словом на знак [карет](#) ^;
- 6) выделить часть фразы после карет ^;
- 7) найти позицию пробела между оставшимися двумя словами;
- 8) выделить левое слово.

	A	B	C	D	E	F
1	Фраза	Сжать пробелы	Слов до предпоследнего	Заменить пробел перед предпол. словом на карет	Оставить текст после карет	Ответ
2	hi you airport id	hi you airport id		2 hi you^airport id	airport id	airport
3	web do thousands museum	web do thousands museum		2 web do^thousands museum	thousands museum	thousands
4	second house a do ball	second house a do ball		3 second house a^do ball	do ball	do
5		B4: =СЖПРОБЕЛЫ(A2)				
6		C4: =ДЛСТР(B2)-ДЛСТР(ПОДСТАВИТЬ(B2;" ";""))-1				
7		D4: =ПОДСТАВИТЬ(B2;" ";"^";C2)				
8		E4: =ПСТР(D2;НАЙТИ("^";D2)+1;50)				
9		F4: =ЛЕВСИМВ(E2;НАЙТИ(" ";E2)-1)				

Рис. 11. Найти предпоследнее слово в пять шагов

Поскольку часто хотят получить ответ в одной ячейке, указанные выше пять формул можно объединить в одну. Для этого начните с формулы в ячейке F2:

=ЛЕВСИМВ(E2;НАЙТИ(" ";E2)-1)

Вместо ссылки на E2 дважды подставьте формулу из E2 (без знака равно). Получится:

=ЛЕВСИМВ(ПСТР(D2;НАЙТИ("^";D2)+1;50);НАЙТИ(" ";ПСТР(D2;НАЙТИ("^";D2)+1;50))-1)

Теперь выделите формулу в D2 и подставьте ее содержимое четыре раза вместо ссылки на D2. Если вы дойдете до конца, то получите формулу, ссылающуюся только на A2:

The screenshot shows the formula bar for cell B2 containing the following formula: `=ЛЕВСИМВ(ПСТР(ПОДСТАВИТЬ(СЖПРОБЕЛЫ(A2);" ";"^";ДЛСТР(СЖПРОБЕЛЫ(A2))-ДЛСТР(ПОДСТАВИТЬ(СЖПРОБЕЛЫ(A2);" ";""))-1;НАЙТИ("^";ПОДСТАВИТЬ(СЖПРОБЕЛЫ(A2);" ";"^";ДЛСТР(СЖПРОБЕЛЫ(A2))-ДЛСТР(ПОДСТАВИТЬ(СЖПРОБЕЛЫ(A2);" ";""))-1)+1;50);НАЙТИ(" ";ПСТР(ПОДСТАВИТЬ(СЖПРОБЕЛЫ(A2);" ";"^";ДЛСТР(СЖПРОБЕЛЫ(A2))-ДЛСТР(ПОДСТАВИТЬ(СЖПРОБЕЛЫ(A2);" ";""))-1;НАЙТИ("^";ПОДСТАВИТЬ(СЖПРОБЕЛЫ(A2);" ";"^";ДЛСТР(СЖПРОБЕЛЫ(A2))-ДЛСТР(ПОДСТАВИТЬ(СЖПРОБЕЛЫ(A2);" ";""))-1)+1;50))-1)`

	A	B	C	D	E	F	G	H
1	Фраза	Ответ						
2	hi you airport id	airport						
3	web do thousands museum	thousands						
4	second house a do ball	do						

Рис. 12. Объединенная формула в одной ячейке

Вот именно в таких задачах выгода от использования функции LET становится очевидной. Для начала вы дадите имя переменной для формулы из B2 (см. рис. 11), затем, используя это первое имя, поименуете формулу в ячейке C3, и т.д., двигаясь теперь слева направо. Вы можете провести подготовительную работу в таблице...

Имя	Значение имени
СжПр	СЖПРОБЕЛЫ(A2)
ЧисСл	ДЛСТР(СжПр)-ДЛСТР(ПОДСТАВИТЬ(СжПр;" ";""))-1
Карет	ПОДСТАВИТЬ(СжПр;" ";"^";ЧисСл)
ДваПосл	ПСТР(Карет;НАЙТИ("^";Карет)+1;30)
	Вычисление
	ЛЕВСИМВ(ДваПосл;НАЙТИ(" ";ДваПосл)-1)

Рис. 13. Имена и значения

... а затем подставить эти имена и формулы в функцию LET:




Рис. 14. Функция LET облегчает написание и понимание формулы

Чтобы формула с LET лучше читалась, используйте Alt+Enter для перехода к новой строке в строке формул после каждой пары аргументов.

Расчет ответа в игре Быки и коровы

Кратко напомню правила. Играют двое. Каждый задумывает и записывает тайное 4-значное число с неповторяющимися цифрами (ноль может быть первым). Игрок, который начинает игру по жребию, делает попытку отгадать число. Попытка — это 4-значное число с неповторяющимися цифрами, сообщаемое противнику в виде вопроса. Противник говорит в ответ, сколько цифр угадано с совпадением их позиций в тайном числе и сколько угадано без совпадения. Например: задумано тайное число 3219; попытка (вопрос) 2310; результат (ответ): один «бык» (цифра 1 из вопроса входит в тайное число и стоит на своем месте) и две «коровы» (цифры 2 и 3 из вопроса входят в тайное число, но стоят не на своем месте). Ответ сообщается в виде 2-значного числа. В нашем примере ответ – 12 (один «бык», две «коровы»). Игроки делают попытки по очереди. Побеждает тот, кто первым получит на свой вопрос ответ 40.

Около 10 лет назад я написал [заметку](#) на эту тему и для нахождения ответов использовал довольно сложную формулу:

Рис. 15. Формула для получения ответа в игре Быки и коровы

Тогда я использовал механизм объединения нескольких формул в ряде ячеек в одну сложную формулу. Промежуточные вычисления у меня не сохранились, так что сейчас мне было бы довольно непросто понять, как работает формула. За 10 лет изменился Excel и мои знания формул, так что сегодня эту задачу я бы решал иначе.

Итак, нам нужно сравнить два 4-значных числа, и подсчитать, сколько цифр совпадает в них с точностью до позиции и сколько просто совпадает:

	A	B	C	D	E	F	G	H	I	J
1	Число	Счет		Вопрос1	2791					
2	0123	02								
3	0124	02				Имя	Значение имени			
4	0125	02				Перв	=ЛЕВСИМВ(A2)	0		
5	0126	02				Втор	=ЛЕВСИМВ(ПРАВСИМВ(A2;3))	1		
6	0127	03				Трет	=ЛЕВСИМВ(ПРАВСИМВ(A2;2))	2		
7	0128	02				Четв	=ПРАВСИМВ(A2)	3		
8	0129	03				ПервВ	=ЛЕВСИМВ(\$E\$1)	2		
9	0132	02				ВторВ	=ЛЕВСИМВ(ПРАВСИМВ(\$E\$1;3))	7		
10	0134	01				ТретВ	=ЛЕВСИМВ(ПРАВСИМВ(\$E\$1;2))	9		
11	0135	01				ЧетвВ	=ПРАВСИМВ(\$E\$1)	1		
12	0136	01				Бык	СУММ(Перв=ПервВ;Втор=ВторВ;Трет=ТретВ;Четв=ЧетвВ)			
13	0137	02				Кор1	СУММ(Перв=ВторВ;Перв=ТретВ;Перв=ЧетвВ)			
14	0138	01				Кор2	СУММ(Втор=ПервВ;Втор=ТретВ;Втор=ЧетвВ)			
15	0139	02				Кор3	СУММ(Трет=ПервВ;Трет=ВторВ;Трет=ЧетвВ)			
16	0142	02				Кор4	СУММ(Четв=ПервВ;Четв=ВторВ;Четв=ТретВ)			
17	0143	01					Вычисление			
18	0145	01					=Бык&(Кор1+Кор2+Кор3+Кор4)			
19	0146	01								

Рис. 16. Формула для получения ответа в игре Быки и коровы с использованием LET

Первые четыре переменные извлекают цифры из проверяемого числа. Вторые четыре переменные – из числа-вопроса. Имя *Бык* возвращает количество совпадающих цифр с точностью до позиции. Кор1, ... Кор4 возвращают число совпадений на разных позициях. *Вычисление* выводит ответ. Вуаля!

Литература

Baxter Floyd. Excel 2021. A Complete Guide About Excel 2021 With All-in-One Approach For Beginners

Bill Jelen. MrExcel 2021 – Unmasking Excel. Holy Macro! Books. 2021

[Разработка](#) оптимальной стратегии игры «Быки и коровы» на основе теории информации